
Length of meridian arc on an ellipsoid 

Suppose the Earth is modelled as an ellipsoid with semi-major axis a, semi-minor axis b, and eccentricity e. A standard problem is to 

find the length of a meridian arc between two geographical latitudes, say φ  and φ . Three methods are suggested below. For more on 

this topic see e.g. the lecture notes by Deakin and Hunter [1].  

The radius of curvature in meridian at latitude φ is given by  

where  ε  =  e /(1 − e )  =  (a  − b )/b . 
 

We work here in terms of ε, which is assumed to be small (ε ≈ 1/150 for the Earth). The arc length is given by  

Method 1: Numerical integration 

In theory this method is subject to rounding errors, due to the addition of many terms. In practice it seems to work well enough, 

provided one uses an accurate integration formula such as Weddle’s rule (rather than the better known Simpson’s and three-eighths 

rules). In Weddle’s rule the number of intervals is a multiple of 6, and the integral over a block of 6 intervals is estimated as  

The number of intervals can be repeatedly doubled until the value stays constant within the desired order of accuracy. This method is 

slow, but is useful as a check on other methods.  

Method 2: integration of the power series 

Here we expand the integrand  (1 + εcos φ)   by the binomial theorem, taking as many terms as are required to achieve the desired 

accuracy. The powers  cos φ  are integrated by rewriting them in terms of multiple angles:  

We may as well multiply the result by the binomial expansion of (1 + ε) , since this does not make it any more complicated. If this is 

done, the length of the meridian arc is  

where the  c   are power series in  ε. To reduce the effect of rounding errors when  φ  − φ   is small, it may be better to rewrite 

 sin2mφ  − sin2mφ   as  2sinm(φ  − φ )cosm(φ  + φ ). The coefficients  c   are given below up to  ε   (equivalent to e ), which should 

be more than enough for most purposes.  

ρ = a(1 − e )(1 − e sin φ) = a(1 + ε) (1 + εcos φ)

Method 3: Another power series method 

The following method may be of interest in that it can be extended by computer to any desired degree of accuracy, without the need to 

program-in complicated coefficients like those in Method 2. It finds the arc length along the meridian from the equator to a given 

latitude φ. The arc length from φ  to φ  can be found by applying the method to φ  and φ , and taking the difference.  

We expand  (1 + εcos φ)   by the binomial theorem, as in Method 2, but now integrate the  cos φ  as follows. Setting 
 

we find easily that  2mK  = (2m − 1)K  + sinφcos φ, whence 
 

A routine calculation shows that the integral, which has to be multiplied by a(1 + ε)   (= a /b)  to get the arc length, is 
 

It is clear how the pattern can be continued indefinitely. The following tested piece of code, written in Delphi Pascal, shows how the 

method can be applied to give any degree of precision up to the limit of the language used.  

const 
  MAX_POWER_EPSILON = 8;  // arbitrary: higher if machine precision warrants it 
var 
  coeff : array [0..MAX_POWER_EPSILON] of extended; 
  // Delphi "extended" is floating-point with 64-bit precision (19 sig. fig.) 
 
{--------------------------------------------------------------------------------------------------- 
Compute the coefficients of phi, sin(phi)*cos(phi), sin(phi)*cos^3(phi), etc. 
This procedure need only be called once for a given ellipsoid (a, b). 
} 
procedure ComputeCoefficients( a, b : extended); 
var 
  epsilon : extended; 
  j, k : integer; 
  denom, mult : extended; 
begin 
  epsilon := (a - b)*(a + b) / (b*b); 
 
  // Compute sums shown inside large () in Web text 
  j := MAX_POWER_EPSILON; 
  denom := 4.0*j*j; 
  coeff[j] := (denom - 1.0)/denom; 
  for j := MAX_POWER_EPSILON - 1 downto 0 do begin 
    for k := MAX_POWER_EPSILON downto j + 1 do coeff[k] := -epsilon * coeff[k]; 
    if (j > 0) then begin 
      denom := 4.0*j*j; 
      coeff[j] := (coeff[j+1] + 1.0) * (denom - 1.0)/denom; 
      end 
    else coeff[0] := coeff[1] + 1.0; 
  end; 

  // Apply factors 1/2, 1.3/2.4, 1.3.5/2.4.6, etc. and dimensions of ellipsoid. 
  mult := a*a/b; 
  coeff[0] := mult*coeff[0]; 
  coeff[1] := mult*coeff[1]; 
  for j := 2 to MAX_POWER_EPSILON do begin 
    mult := mult*(2*j - 3)/(2*j - 2); 
    coeff[j] := mult*coeff[j]; 
  end; 
end; 
 
{--------------------------------------------------------------------------------------------------- 
Apply the above coefficients to compute arc length on meridian 
  from equator to passed-in latitude phi. 
} 
function ComputeArcLengthFromEquator( phi_degrees : extended) : extended; 
var 
  phi, c, c2, total : extended; 
  j : integer; 
begin 
  phi := (PI/180.0) * phi_degrees;  // degrees to radians 
  c := Cos( phi); 
  c2 := c*c; 
  total := coeff[MAX_POWER_EPSILON]; 
  for j := MAX_POWER_EPSILON - 1 downto 1 do total := total*c2 + coeff[j]; 
  result := phi*coeff[0] + Sin(phi)*c*total; 
end; 

